

Let the Data Drive it and Set your Hands Free - A Macro to Create Indicators for Special Interest AEs

Yiwei Liu, Jameson Cai, Eli Lilly and Company

ABSTRACT

Adverse Events of Special Interest (AESI) form an important part of the safety profile for a compound in submission or clinical study report (CSR). AESI identification is sophisticated because its information originates from MedDRA, Standardized MedDRA Queries (SMQs), Customized Queries (CQs), and/or Adjudicated Events performed by an independent Adjudication Committee (AC). Semi-annual MedDRA updates further complicate the reporting process. As AESIs are defined at a compound level, reporting AESIs should be consistent across all clinical trials of the compound.

A macro is developed to create a set of ADaM variables that identify multiple types of AESI to streamline the process. It can be executed to automatically create all AESI tabulation and analysis reporting for multiple studies with a common definition. This method not only simplifies and standardizes the AESI reporting process, but also increases accuracy in AESI classification. The macro is portable to all compounds with a similar need.

INTRODUCTION

Adverse Events of Special Interest (AESI) is a critical topic for drug submission and approval. By monitoring and reporting AESIs, researchers can identify potential safety issues and take appropriate actions to protect study participants, meet regulatory requirements, stop the trial when necessary, and ensure the validity of study results. In a clinical trial, AESIs may be identified by the sponsor or regulatory agency and closely monitored and reported using SMQs and/or CQs. SMQs are useful in identifying safety signals for well-known clusters of adverse event terms, while CQs can be tailored to identify specific adverse events that may be of interest in a trial. In addition, events in customized queries may be adjudicated to ensure accurate, reliable, and credible results. Adjudication provides an independent review of the data using a common definition and standard of evidence, which can help to ensure the validity of the trial results and provide valuable information for regulatory authorities, clinicians, and patients.

All the SMQs and CQs submitted need to be indicated in ADAE. As AESI are usually identified on a compound level, reporting AESI has to be consistent across all the clinical trials for the same compound. Seeing the necessity and challenges to report AESI accurately, efficiently, and consistently, we developed a process that automatically identifies and reports AESIs across all studies with the same definition.

This paper introduces a SAS® macro that creates SMQs and CQs analysis variables to meet different needs of AESI reporting in ADAE. The macro provides standardized derivation rules and also improves flexibility to tailor queries for specific study needs.

METHOD:

The first step is to create an Excel look-up table named aesi_smq_cq.xlsx (Table 1). This table contains the variable names, SMQ codes, and specifications for the required variables. The look-up table is flexible and can be updated as per the requirements of AESI.

Once the look-up table is created, it is imported into a SAS dataset. The imported data is then combined with the input Adverse Events (AE) dataset to create the required analysis variables.

The use of a look-up table allows for easy modification of the macro and ensures that the analysis variables are consistent with the current requirements for AESI reporting. By importing the look-up table into SAS and combining it with the AE dataset, the macro can be used to create the required analysis variables efficiently and accurately:

SMQID	SMQ	Notes
SMQ01	20000008	For Liver related investigations, signs and symptoms
SMQ02	20000009	For Cholestasis and jaundice of hepatic origin
SMQ03	20000010	For Hepatitis non-infections
SMQ04	20000013	For Hepatic failure, fibrosis and cirrhosis and other liver damage

Table 1. aesi_smq_cq.xlsx

```
%aesi_smq_cq
(inds= , input dataset name
 outds=, output dataset name
 l_sdtm=, library name of SDTM
 l_dict=, library name of meddra_smq dictionary
 tblpath=, path to the lookup tables)

data ae_in;
  set &inds;
run;

proc import datafile="&tblpath./aesi_smq_cq.xlsx"
  out=smq_cq_table dbms=xlsx replace;
  sheet=Sheet1;
run;
```

After combining the Adverse Events (AE) dataset with the SMQ look-up table containing the SMQ code, the next step is to merge it with the MedDRA SMQ dictionary. This is achieved by matching the values of AE.AERPVER, AE.SMQCD, and AE.AEDECOD with the corresponding fields DICTVER, SMQ_CODE, PTERM in the Dict_MedDRA_SMQ dataset (Figure 1).

	DICTVER	SMQ_CODE	SMQ	PTERM	SCOPE	CATEGORY
1	25.1	20000008	Liver related investigations, signs and symptoms (SMQ)	5nucleotidase increased	Broad	A
2	25.1	20000008	Liver related investigations, signs and symptoms (SMQ)	AST to platelet ratio index increased	Broad	A
3	25.1	20000008	Liver related investigations, signs and symptoms (SMQ)	AST/ALT ratio abnormal	Narrow	A
4	25.1	20000008	Liver related investigations, signs and symptoms (SMQ)	Acquired factor V deficiency	Narrow	A
5	25.1	20000008	Liver related investigations, signs and symptoms (SMQ)	Alanine aminotransferase abnormal	Narrow	A
6	25.1	20000008	Liver related investigations, signs and symptoms (SMQ)	Alanine aminotransferase increased	Narrow	A
7	25.1	20000008	Liver related investigations, signs and symptoms (SMQ)	Ammonia abnormal	Narrow	A
8	25.1	20000008	Liver related investigations, signs and symptoms (SMQ)	Ammonia increased	Narrow	A
9	25.1	20000008	Liver related investigations, signs and symptoms (SMQ)	Anti-liver cytosol antibody type 1 positive	Narrow	A
10	25.1	20000008	Liver related investigations, signs and symptoms (SMQ)	Ascites	Narrow	A

Figure 1. Dict_MedDRA_SMQ

If there are any matching records in both datasets, the macro sets the values of the SMQxxNAM, SMQxxCD, and SMQxxSC variables to the values of Dict_MedDRA_SMQ.SMQ, Dict_MedDRA_SMQ.SMQ_CODE and Dict_MedDRA_SMQ.SCOPE, respectively. This ensures that the SMQ name, code, and scope information are correctly identified and associated with the Adverse Events data. See Figure 2.

Once the input AE data and Dict_MedDRA_SMQ datasets are merged, the macro can accurately identify and categorize Adverse Events based on SMQs. This approach helps ensure consistency and accuracy in AESI reporting across multiple studies and reduces the likelihood of errors.

%smq is used to derive SMQ related variables. Code is as follows:

```
%macro smq(indt=, outdt=) ;
```

```

%do n=1 %to 4; *4 is the largest number of SMQ variables, can be modified based on
the requirement ;
%let num = %sysfunc(putn(&n, z2.));
data smq_1;
  set smq_cq;
  where smqid="SMQ&num.";
  smqcd=input(smq, 8.);
run;

*combine SMQ lookup table and input AE data ;
data _adae_in;
  if _n_=1 then set smq_1(keep=smqid smqcd);
  %if &n.=1 %then set &indt.;%else set &outdt.;;
run;

data dict_smq;
  set dict.dict_meddra_smq;
  where pterm ne '';
run;

proc sql;
  create table smq_out as
  select distinct a.*, b.smq, b.scope, b.category
  from _adae_in as a left join dict_smq as b on
  a.smqcd = b.smq_code and a.aedecod = b.pterm and input(b.dictver,best.)
  eq input(a.aerpver,best.);
quit ;

data &outdt.;
  length smq&num.nam $200. smq&num.sc $6.;
  set smq_out ;
  if smq ne '' then do;
    smq&num.nam = smq ;
    smq&num.sc = upcase(scope) ;
    smq&num.cd = smqcd;
  end;
* reset SMQxxCD to be null if not populated ;
  if smq&num.nam = '' then smq&num.cd = . ;
  drop smq smqcd smqid scope category;
run ;

%end;
%mend smq;

%smq(indt=_ae_in, outdt=_ae_smq) ;

```

	SUBJID	SMQ01NAM	SMQ01SC	SMQ01CD	AEDECOD
1	10001				White blood cell count decreased
2	10001	Liver related investigations, signs and symptoms (SMQ)	BROAD	20000008	Blood alkaline phosphatase increased
3	10001	Liver related investigations, signs and symptoms (SMQ)	NARROW	20000008	Gamma-glutamyltransferase increased
4	10001	Liver related investigations, signs and symptoms (SMQ)	NARROW	20000008	Alanine aminotransferase increased

Figure 2. Example of SMQxxNAM, SMQxxSC and SMQxxCD variables created by the macro %SMQ

Besides creating SMQ variables, the macro also derives CQs variables according to medical requirements. Additional adjudicated CQ variables are created in this macro as well. Adjudication information is stored in SDTM FA and linked to AE in RELREC by FAGRPID and AEGRPID. Figure 3 shows an example of the SDTM annotations on the “Adjudication - Cardiac Ischemic Event” case report form (CRF) page. There are two adjudication assessments for Cardiac Ischemic Event: “Myocardial Infarction” and “Hospitalization for Unstable Angina”. If an event meets the event definition, the AC will mark it as Event, otherwise, it will be marked as “No Event” or “Insufficient Documentation for Event Determination”. It is complicated to code this process and consider the adjudication result and different

subcategories. The macro can provide a standard output and handle this process to avoid redundant coding in different studies.

Adjudication - Cardiac Ischemic Event [igAJCIE3001_F1_S1]	
1. Sequence Identifier (Sequence Identifier)	[FASPID_AJCIE_F1] FAOAJ = AEDECOD
2. AE Group ID (AE Group ID)	[AEGRPID_RELREC_AJCIE_F1] RELREC between FA and AE
3. Date of Event (Investigator's Assessment) (Date of Event Investigator's Opinion)	[EPAEDAT_AJCIE_F1] (DD/MM/YYYY) FADTC
4. Does Adjudicator's assessment of the Event Date agree with Primary Investigator's assessment of the Event Date? (Investgr and Adjctr Event Date Agree)	[EVDTAGMT_AJCIE_F1] <input type="radio"/> Yes <input type="radio"/> [AJAEDAT_AJCIE_F1] (DD/MM/YYYY) No Date of Event (Adjudicator's Assessment) <input type="radio"/> / / <input type="radio"/> / / <input type="radio"/> Not Applicable FAORRES when FATESTCD = EVDTAGMT
5. CEC Adjudication Assessment (CEC Adjudication Assessment)	[CECAJASM_AJCIE_F1] <input type="radio"/> [CIER1_AJCIE_F1] Event Cardiac Ischemic Event Type <input type="radio"/> [CIER2_AJCIE_F1] Myocardial Infarction Myocardial Infarction Classification <input type="radio"/> Spontaneous Clinically Overt MI related to ischemia due to a primary coronary event (Type 1) <input type="radio"/> MI secondary to ischemia due to either increased oxygen demand or decreased supply (Type 2) <input type="radio"/> Sudden Unexpected Cardiac Death, often with symptoms suggestive of myocardial ischemia (Type 3) <input type="radio"/> Type 4b MI associated with stent thrombosis as documented by angiography or at autopsy <input type="radio"/> Type 4a Percutaneous Coronary Intervention-Related MI <input type="radio"/> Type 4c Stent restenosis without occlusion <input type="radio"/> Coronary Artery Bypass Grafting-Related MI (Type 5) <input type="radio"/> Unknown <input type="radio"/> Hospitalization for Unstable Angina <input type="radio"/> No Event <input type="radio"/> Insufficient Documentation for Event Determination FAORRES when FATESTCD = CECAJASM FAORRES when FATESTCD = CIER1 FAORRES when FATESTCD = CIER2
6. Category for Findings About (Category for Findings About)	[FACAT_AJCIE_F1] <input type="radio"/> Adjudication - Cardiac Ischemic Event FACAT
7. Evaluator (Evaluator)	[AJCIEEVAL_AJCIE_F1] <input type="radio"/> Adjudication Committee FAEVAL

Figure 3. Adjudication - Cardiac Ischemic Event CRF

This macro derives two types of adjudication variables based on the above information: CQxxSCAx (subcategory, for adjudicated names) and CQxxTYPx (report type: "Event", "No Event" or "Insufficient Documentation for Event Determination").

Figure 4 displays the FA input and output data for the macro, which includes information about the adjudication of events for different subjects. Specifically, for subject 1001, the event "Myocardial Infarction" was approved by the adjudication committee (AC) in the category of "ADJUDICATION - CARDIAC ISCHEMIC EVENT" (FACAT), with two corresponding records in the dataset. The first record (FAORRES="MYOCARDIAL INFARCTION") indicates that the event was identified as "Myocardial Infarction" when the value of FATESTCD was "CIER1". The second record (FAORRES= EVENT) indicates that the event type was classified as "Event" when the value of FATESTCD was "CECAJASM" in FA. Therefore, CQ06SC2 corresponds to "Myocardial Infarction" and CQ06TYP2 corresponds to "Event" for this subject in output data.

Similarly, for subject 1003, the event "Coronary artery disease" was adjudicated as "Hospitalization for Unstable Angina", so CQ06SC3 equals to "Hospitalization for Unstable Angina" and CQ06TYP2 equals "Event".

For subject 1002, the event "Chest discomfort" did not meet the "Cardiac Ischemic Event" (FACAT) definition and was therefore not approved by the AC (FAORRES="NO EVENT"). Consequently, the values of CQ06TYP2 and CQ06TYP3 for this subject were both set to "No Event", corresponding to CQ06SC2 and CQ06SC3, respectively, which are different from the corresponding values for subjects 1001 and 1003.

Similar adjudication categories can be derived for other events based on the information provided in the CRFs.

```

data _rel_fa;
  merge fa _relrec_ae (in=a); *merge Relrec data with FA by fagrpid;
  by usubjid subjid fagrpid;
  if a;
  length value $200;
  if upcase(facat) = "ADJUDICATION - CARDIAC ISCHEMIC EVENT" then do;
* Derive adjudicated Event variables for "Myocardial Infarction";
  if FATESTCD="CIER1" and upcase(faorres) = "MYOCARDIAL INFARCTION" then do;
    id = "CQ06SCA2"; value = "Myocardial Infarction"; output;
    id = "CQ06TYP2"; value = "Event"; output;
  end;
* Derive adjudicated Event variables for "Hospitalization for Unstable Angina";

```

```

else if FATESTCD="CIER1" and upcase(FAORRES)="HOSPITALIZATION FOR UNSTABLE ANGINA" then
do;
    id = "CQ06SCA3"; value = "Hospitalization for Unstable Angina"; output;
    id = 'CQ06TYP3'; value = 'Event';output;
end;
*Derive adjudicated Non Event or Insufficient documentation for event determination records for
both "Myocardial Infarction" and "Hospitalization for Unstable Angina";
else if FATESTCD='CECAJASM' and (upcase(FAORRES)="NO EVENT" or
upcase(FAORRES)="INSUFFICIENT DOCUMENTATION FOR EVENT DETERMINATION") then do;
    id = "CQ06SCA2"; value = "Myocardial Infarction"; output;
    id = 'CQ06TYP2'; value = propcase(FAORRES);output;
    id = "CQ06SCA3"; value = "Hospitalization for Unstable Angina"; output;
    id = 'CQ06TYP3'; value = propcase(FAORRES);output;
end;
end;
run;

proc sort data=_rel_fal nodupkey;
by subjid aegrpid id value;
run;

proc transpose data=_rel_fal out=_rel_fa_t;
by subjid aegrpid;
id id;
var value;
run;

```




Figure 4. Input and output for Adjudicated CQ variables: CQ06SCA2, CQ06TYP2, CQ06SCA3 and CQ06TYP3

More additional CQs variables can be derived from this macro based on the AESI requirements, such as hepatic events, infection, depression, and malignancies etc.

CONCLUSION

With all the tools as input, these component modules can be executed to automatically create AESI tabulation and analysis reporting for each study with the same AESI definition. As a result, the process not only simplifies and standardizes AESI reporting, but also enables faster and more accurate AESI reporting for the compound. The process can be adopted for other compounds as well.

ACKNOWLEDGMENTS

The authors would like to express their gratitude to Lorena Hernandez Maxwell for creating the lookup table, as well as to Janice Yan for her encouragement and support.

CONTACT INFORMATION

Your comments and questions are valued and encouraged. Contact the authors at:

Yiwei Liu
Eli Lilly and Company
Email: liu_yiwei@lilly.com

Jameson Cai
Eli Lilly and Company
Email: jameson.cai@lilly.com

Any brand and product names are trademarks of their respective companies.